Specifying Required and Default Parameters

If you pass more arguments to a macro than there are parameters, the additional arguments generate a warning (unless you use the VARARG keyword). If you pass fewer arguments than the macro procedure expects, the assembler assigns empty strings to the remaining parameters (unless you have specified default values). This may cause errors.

MASM provides the ability to specify that certain macro parameters are required. You can also assign a default value to a parameter if the programming doesn't supply one. Finally, MASM also provides facilities to allow a variable number of macro arguments.
The syntax for a required parameter is:
parameter:REQ

At assembly time, MASM will generate an error if a required parameter is missing.

Example:

WRITECHAR  MACRO CHAR:REQ

    PUSH AX

    PUSH DX

    MOV  DL, CHAR               
;; Select ASCII char
    MOV  AH, 02H                 
 
;; Select DOS Print Char function
    INT  21H                   

 ;; Call DOS

    POP DX

    POP AX
ENDM

The syntax for a parameter with default value is:

 parameter:=<textValue>

Example:

WRITECHAR  MACRO char:=<07H>

     PUSH AX

    PUSH DX

    MOV  DL, CHAR               
;; Select ASCII char
    MOV  AH, 02 H                
 ;; Select DOS Print Char function
    INT  21H                    
;; Call DOS

    POP DX

    POP AX
ENDM

If a reference to this macro does not include the argument char, the assembler fills in the blank with the default value of 07H and the macro beeps when called.
MASM can process a variable number of macro parameters. To do this, the last formal parameter is suffixed with :VARARG to indicate that a single named parameter will have the actual value of all additional arguments.

For example, the following macro definition includes the three possible parameter attributes — required, default, and variable:
work    MACRO   arg1:REQ, arg2:=<5>, arg3:VARARG

The variable argument must always be last. If this macro is called with the statement
        work 4, , 6, 7, a, b

the first argument is received as the value 4, the second is replaced by the default value 5, and the last four are received as the single argument <6, 7, a, b>.

The for macro can be used to extract each parameter from this variable argument list. The for macro is explained later.

Macro Operators

The macro operators that MASM provides are:
	Symbol
	Name
	 Description

	 ;;
	 Suppress-comment 

  operator
	

	 &
	 Substitution Operator
	 Tells the assembler to replace a macro 

 parameter name with its actual value.

	 < >
	 Text Delimiters
	 Opens and closes a literal string.

	 !
	 Literal-Character

 Operator
	 Treats the next character as a literal character, 

 even if it would normally have another

  meaning. 

	 %
	 Expansion Operator
	 Causes the assembler to expand a constant

  expression.


 

The Suppress-comment operator ;;
This operator suppresses comments from appearing in macro expansions.

Substitution Operator &

References to a parameter within a macro can sometimes be ambiguous. In such cases, the assembler may not expand the argument as you intend. The substitution operator (&) lets you identify unambiguously any parameter within a macro.
errgen  MACRO   num, msg
    err&num BYTE  “Error &num: ”, msg, “$”
ENDM

When called with the following arguments,
errgen  5, “Unreadable disk”

the macro now generates this data definition:
err5    BYTE    “Error 5: ”, “Unreadable disk”, “$”

When it encounters the & operator, the assembler interprets subsequent text as a parameter name until the next & or until the next separator character (such as a space, tab, or comma). The expression err&num could also be written as err&num& .
The rule also works in reverse. You can delimit a parameter reference with & at the end rather than at the beginning. For example, if num is 5, the expression num&12 resolves to 512
The assembler processes substitution operators from left to right. For example, if arg1 has the value var and arg2 has the value 3, you could paste them together with this statement:
&arg1&&arg2&    BYTE    “Text”

or 

arg1&&arg2      BYTE    “Text”

this generates the data definition:


var3    BYTE  “Text”

Example: Write a macro that sorts two 16-bit signed, or unsigned memory operands, based on a condition passed as argument

SORT2  MACRO  VAR1_16, CONDTN, VAR2_16


LOCAL  DONE


PUSH  AX


MOV  AX,  VAR1_16


CMP  AX,  VAR2_16


J&CNDTN  DONE


XCHG  AX,  VAR2_16


MOV  VAR2_16,  AX

     DONE:


POP  AX

ENDM
The call:

    SORT2   VAR1, BE, VAR2

will sort the 16-bit variables VAR1 and VAR2 in decreasing order.

Text Delimiters

The angle brackets (< >) are text delimiters. 
By delimiting the text of macro arguments, you can pass text that includes spaces, commas, semicolons, and other special characters. The following example expands a macro called work in two different ways:
   work    <1, 2, 3, 4, 5> 
     ; Passes one argument with 13 chars, including commas and spaces


   work    1, 2, 3, 4, 5   
      ; Passes five arguments, each with 1 character

Text delimiters have a special use with the FOR macro, as explained later on.

The Literal-Character Operator

The literal-character operator (!) lets you include special characters: ! & > % as part of text delimited with < >. The assembler treats the character following ! literally rather than as a special character.

Notice the following differences between text delimited with < > and that delimited with single or double quotes when used as macro argument:

1. A < > delimited string can be inserted within another string within the macro.

2. A < > delimited string requires the literal-character operator(!) before each special character.

       ALLOCATE1   MACRO   TEXTNAME,  TEXT


 TEXTNAME    BYTE   TEXT

       ENDM

        ALLOCATE2   MACRO   TEXTNAME,  TEXT


 TEXTNAME    BYTE   ‘&TEXT’

       ENDM

      errgen1  MACRO   num, msg
    err&num BYTE  “Error &num: ”, msg, “$”
   ENDM
   errgen2  MACRO   num, msg
    err&num BYTE  “Error &num: &msg&$”
   ENDM
     .DATA

         ALLOCATE1    TEXT1,   “DANGER! TEMPERATURE > 100 DEGREES CENTIGRADE”

         ALLOCATE2    TEXT2,   <DANGER!! TEMPERATURE !> 100 DEGREES CENTIGRADE>

         errgen1  5, “Unreadable disk”

       errgen2  6, <Unformatted disk.>
Expansion Operator

The expansion operator (%) forces immediate evaluation of a constant expression and replaces it with a text value consisting of the digits of the result. The digits are generated in the current radix (the default is decimal).

The expansion operator gives you flexibility when passing arguments to macros. It lets you pass a computed value rather than the literal text of an expression. 

Example:
work    MACRO   arg
    mov ax, arg * 4
ENDM

The macro accepts different arguments:
        work    2 + 3           
; Passes “2 + 3”
                               
   
 ; Code: mov ax, 2 +  3 * 4 
        work    %2 + 3         
 ; Passes 5
                                

; Code: mov ax, 5 * 4


You must consider operator precedence when using the expansion operator. Parentheses inside the macro can force evaluation in a desired order:
work    MACRO   arg
    mov ax, (arg) * 4
ENDM

        work    2 + 3           ; Code: mov ax, (2 + 3) * 4
        work    %2 + 3          ; Code: mov ax, (5) * 4

-------------------------------------------------------------------------------

Defining Repeat Blocks with Loop Directives

A “repeat block” is an unnamed macro defined with a loop directive. The loop directive generates the statements inside the repeat block a specified number of times or until a given condition becomes true.
Repeat blocks can be used outside macros, but they frequently appear inside macro definitions to perform some repeated operation in the macro. Since repeat blocks are macros themselves, they end with the ENDM directive.
MASM supports the following four loop directives: REPEAT, WHILE, FOR, and FORC. In versions of MASM prior to 6.0, REPEAT was called REPT, FOR was called IRP  (Indefinite RePeat), and FORC was called IRPC (Indefinite RePeat Character). MASM 6.1 recognizes the old names.
REPEAT Directive

•The REPEAT directive repeats a statement block a fixed number of times.

•Syntax:

REPEAT constExpression


statements

ENDM

Where ConstExpression is an unsigned constant integer expression. It determines the number of repetitions. The expression must contain no forward references.

Here is an example of a repeat block used to generate data. It initializes an array containing sequential ASCII values for all uppercase letters.

alpha   LABEL   BYTE            ;  Name the data generated
letter  =       ‘A’            
 ;  Initialize counter
REPEAT  26                      
;; Repeat for each letter
    BYTE    letter              
;; Allocate ASCII code for letter
    letter  = letter + 1        
;; Increment counter
ENDM

This effectively generates the following statements: 

      alpha     byte            'A'

                byte            'B'

                 .

                 .

                 .

                byte            'Y'

                byte            'Z'

           letter       =       27
Here is another use of REPEAT, this time inside a macro:

beep    MACRO   iter:=<3>

    PUSH AX

    PUSH DX
    mov ah, 2                   ;; Character output function
    mov dl, 7                   ;; Bell character
    REPEAT iter                 ;; Repeat number specified by macro
        int 21h                 ;; Call DOS
    ENDM

    POP DX

    POP AX
ENDM

------------------------------------------------------------------------------

•The WHILE directive repeats a statement block as long as a particular constant expression is true. 

•Syntax:

WHILE constExpression


statements

ENDM

The constExpression must be a value that can be calculated at assembly time. Normally, the expression uses relational operators (EQ, NE, GT, LT, GE, and LE), but it can be any expression that evaluates to zero (false) or nonzero (true). 

The following operators may also appear in the constExpression:

      Arithmetic operators: +, -, *, /, mod, + (unary), - (unary)

      Logical operators: NOT, AND, OR, XOR
The following repeat block uses the WHILE directive to allocate variables initialized to calculated values. This is a common technique for generating lookup tables. 
cubes   LABEL   WORD            ;; Name the data generated
root    =   1                                   ;; Initialize root
cube    =   root * root * root         ;; Calculate first cube
WHILE   cube LE 32767             ;; Repeat until result too large
    WORD   cube                         ;; Allocate cube
    root   =    root + 1                    ;; Calculate next root and cube
    cube   =    root * root * root
ENDM
FOR Directive

• The FOR directive repeats a statement block by iterating over a comma-separated list of

  arguments. The argument list must always be enclosed in angle brackets.

• Each argument in the list causes one iteration of the loop.

• Syntax:
FOR parameter,<arg1,arg2,arg3,. . ., argN>


statements

ENDM
Example:

array label byte 

for  value,<0,1,2,3,4,5>

    byte    value

endm
generates:

     array   byte    0

                byte    1

                byte    2

                byte    3

                byte    4

               byte     5

Example:
series  LABEL   BYTE
FOR arg, <1,2,3,4,5,6,7>
    BYTE  arg DUP (arg)
ENDM

generates:

series  
byte 1
DUP(1)

           
byte 2
DUP(2)


byte 3
DUP(3)


byte 4
DUP(4)


byte 5
DUP(5)


byte 6
DUP(6)


byte 7
DUP(7)

Example:

Window STRUCT


FOR color,<frame,titlebar,background,foreground>



color DWORD ?


ENDM

Window ENDS

generates:

Window STRUCT

   frame DWORD ?

   titlebar DWORD ?

   background DWORD ?

   foreground DWORD ?

Window ENDS

FOR Loops and Variable-Length Parameters

 The FOR directive also provides a convenient way to process macros with a variable number of arguments. 
The following macro illustrates variable arguments:
show    MACRO chr:VARARG

   PUSH  AX

   PUSH  DX
    mov     ah, 02h
    FOR arg, <chr>
        mov     dl, arg
        int     21h
    ENDM

    POP  DX

    POP  AX
ENDM

When called with
         show  ‘O’, ‘K’, 13, 10

the macro displays each of the specified characters one at a time.

The parameter in a FOR loop can have the required or default attribute. You can modify the show macro to make blank arguments generate errors:  

show    MACRO chr:VARARG

     PUSH  AX

    PUSH  DX
    mov     ah, 02h
    FOR arg:REQ, <chr>
        mov     dl, arg
        int     21h
    ENDM

    POP  DX

    POP  AX
ENDM

The macro now generates an error if called with
        show  ‘O’,, ‘K’, 13, 10

Another approach would be to use a default argument:

show    MACRO chr:VARARG

    PUSH  AX

    PUSH  DX
    mov     ah, 02h
    FOR arg:=<‘ ’>, <chr>
        mov     dl, arg
        int     21h
    ENDM

    POP  DX

    POP  AX
ENDM

Now calling the macro with
        show  ‘O’,, ‘K’, 13, 10

inserts the default character, a space, for the blank argument.

The following macro can be used to PUSH any number of 16- or 32-bit registers passed as arguments:

  PUSHM MACRO REGISTER:VARARG

     FOR ARG:REQ, <REGISTER>

        PUSH ARG

     ENDM

  ENDM

The following macro can be used to POP any number of 16- or 32-bit registers passed as arguments:

POPM MACRO REGISTER:VARARG

     FOR ARG:REQ, <REGISTER>

        POP ARG

     ENDM

  ENDM

Example:

   PUSHM   AX,  BX,  DI,  ECX

    .

    .

    .
   POPM   ECX,  DI,  BX,  AX

FORC Directive

• The FORC directive repeats a statement block by iterating over a string of text (including

   spaces). The text must be enclosed in angle brackets. Each character in the string causes one iteration of the loop.

• Syntax:
FORC parameter, < text>
 statements
ENDM

The following example illustrates FORC:

FORC code,<ABCDEFG>


Group_&code WORD ?

ENDM

 Group_A WORD ?

 Group_B WORD ?

 Group_C WORD ?

 Group_D WORD ?

 Group_E WORD ?

 Group_F WORD ?

 Group_G WORD ?
FORC arg, <ABCDEFGHIJKLMNOPQRSTUVWXYZ>
    BYTE  ‘&arg’             ;; Allocate uppercase letter
    BYTE  ‘&arg’ + 20h       ;; Allocate lowercase letter
    BYTE  ‘&arg’ - 40h       ;; Allocate ordinal of letter
ENDM

Notice that the substitution operator must be used inside the quotation marks to make sure that arg is expanded to a character rather than treated as a literal string.
MACRO FUNCTIONS

A macro function is a named group of statements that returns a value. When calling a macro function, you must enclose its argument list in parentheses, even if the list is empty. The function always returns text. 
You define macro functions in exactly the same way as an ordinary macro, except that a macro function always returns a value through the EXITM directive.
The return value must be text. A macro function must first convert a numeric value before returning it. The macro function can use angle brackets or the expansion operator (%) to convert numbers to text. 

Here is an example of a macro function that uses the WHILE directive to calculate factorials:

factorial   MACRO   num:REQ
    LOCAL   i, factor
    factor  =   num
    i       =   1
    WHILE   factor GT 1
        i       =   i * factor
        factor  =   factor - 1
    ENDM
    EXITM   %i
ENDM

The integer result of the calculation is changed to a text string with the expansion operator (%). The factorial macro could have returned its value as  
         EXITM   <i>

The factorial macro can define data, as shown here:
      var     WORD    factorial( 4 )

This statement initializes var with the number 24 (the factorial of 4).

The following macro function determines the number of arguments in a VARARG parameter:   

hiddenArgCount MACRO arglist:VARARG
    LOCAL count
    count = 0
    FOR arg, <arglist>
        count = count + 1       ;; Count the arguments
    ENDM
    EXITM %count
ENDM

 CONDITIONAL ASSEMBLY

Conditional assembly allows object code generated to be varied at assembly time. Once assembled, the object code is fixed and immutable.

Conditional assembly may be used with macros to make them more powerful.

Simple conditional-assembly directives have the general form:


IF   condition

                   statement(s)

            ENDIF

A directive may contain an ELSE part:


IF  condition

                 statement(s)

            ELSE

                 statement

            ENDIF

The IF directive and its condition may be replaced by any of the other conditional-assembly directives given in the following table:

	IF expression
	Grants assembly if expression is true (nonzero)

	IFE expression


	“IF falsE” grants assembly if expression is false (0). 

	IFB <arg>


	“IF Blank” grants assembly if the text item arg is blank. 

	IFNB < arg>


	“IF NotBlank” grants assembly if the text item arg is not blank.

	IFIDN <arg1>, <arg2>


	“IF IdeNtical” grants assembly if the text items arg1 and arg2 are identical. The comparison is case sensitive.

	IFIDNI <arg1>, <arg2>
	“IF IdeNtical Ignoring case” grants assembly if the text items arg1 and arg2 are identical. The comparison is case insensitive.

	IFDIF <arg1>, <arg2>


	“IF DIFferent” grants assembly if the text items arg1 and arg2 are different. The comparison is case sensitive.

	IFDIFI <arg1>, <arg2>
	“IF DIFferent Ignoring case” grants assembly if the text items arg1 and arg2 are different. The comparison is case insensitive. 

	IFDEF name


	“IF DEFined” grants assembly if name is a previously defined label, variable, or symbol. 

	IFNDEF name


	“IF Not DEFined” grants assembly if name has not been defined.


IFs can be nested, and a corresponding ENDIF is always required with each IF. A special form of nesting is allowed using multiple IF / ELSEIF pairs with one ending ENDIF as shown below:

IF condition1

      statement(s)

ELSEIF condition2

     statement(s)

ELSEIF condition3

     statement(s)

ELSEIF condition4

.

.

.

ELSE (optional)

    statement(s)

ENDIF
The following directives may be substituted for ELSEIF: 

        ELSEIFB, ELSEIFDEF, ELSEIFDIF, ELSEIFDIFI, ELSEIFE, ELSEIFIDN, 

        ELSEIFIDNI, ELSEIFNB, and ELSEIFNDEF.

The condition for IF and ELSEIF may contain any of the following operators:

           Arithmetic operator:  +, -, *, /, mod, + (unary), - (unary)

           Relational operators:  EQ, NE, GT, LT, GE, LE

           Logical operators:  NOT, AND, OR, XOR

            TYPE operator: It returns the number of bytes reserved in memory for an

           operand:

	Operand
	Number of bytes

	BYTE , SBYTE
	1

	WORD, SWORD
	2

	DWORD, SDWORD
	4

	QWORD
	8

	constant
	0


     Only assembly-time constants may be compared using these operators.
The following are useful macro directives:

	ECHO message
	The echo directive simply displays whatever appears in its operand field to the video display during assembly It is the same as: 

   %OUT  message

	EXITM


	The EXITM directive immediately terminates the expansion of a macro, exactly as though MASM encountered ENDM. MASM ignores all text from the EXITM directive to the ENDM.


Examples:

The following macro displays an error message at assembly time if the string argument is missing:

mWriteStr MACRO string

   IFB <string>

     ECHO -----------------------------------------

     ECHO * Error: parameter missing in mWriteStr

     ECHO * (no code generated)

     ECHO -----------------------------------------

     EXITM

   ENDIF

   PUSH AX

   PUSH DX

   MOV AH, 09H

   MOV DX,OFFSET string

   INT 21H  

   POP DX

   POP AX

ENDM
-------------------------------------------------------------------------------------------------------------------------------------

The following macro displays an eror message at assembly time if the maxChars argument is the EDX register:

mReadBuf MACRO bufferPtr, maxChars

  IFIDNI <maxChars>,<EDX>

    ECHO Warning: Second argument cannot be EDX

    ECHO **************************************

    EXITM

  ENDIF

  . 

  .

  .

ENDM
--------------------------------------------------------------------------------------------------------------------------------------------------------------------

The following macro performs a memory-to-memory MOV on two 8-bit, 16-bit, or 32-bit operands:

MMOVE MACRO DEST, SOURCE

   PUSH EAX

   IF (TYPE DEST) EQ 1 AND (TYPE SOURCE) EQ 1

     MOV AL, SOURCE

     MOV DEST, AL

   ELSEIF (TYPE DEST) EQ 2 AND (TYPE SOURCE) EQ 2

     MOV AX, SOURCE

     MOV DEST, AX

   ELSEIF (TYPE DEST) EQ 4 AND (TYPE SOURCE) EQ 4

     MOV EAX, SOURCE

     MOV DEST, EAX

   ELSE

     ECHO ERROR- INVALID PARAMETERS

   ENDIF

   POP EAX

 ENDM

--------------------------------------------------------------------------------------------------------------------

The following macro displays an immediate string, with or without carriage return and line feed:

 WRITE MACRO TEXT, CRLF

    LOCAL STRING

    PUSHM AX, BX, DX

    MOV AH, 40H

    MOV BX, 01H

    MOV DX, OFFSET STRING

    MOV CX, LENGTHOF STRING

    INT 21H

    POPM DX, BX, AX

    .DATA

       IFB <CRLF>

         STRING BYTE TEXT

       ELSE

         STRING DB TEXT, 0DH, 0AH

       ENDIF

     .CODE

  ENDM
DIRECTIVES CONTROLING THE CONTENTS OF A LIST FILE

The following directives control the contents of a list file:

	Directive
	
	Directive
	

	.LIST


	Starts listing of statements. This is the default.
	.NOLIST


	Suppresses program listing. Same as .XLIST.

	.LISTMACROALL


	Starts listing of all statements in macros. Same as .LALL.
	.NOLISTMACRO


	Suppresses listing of macro expansions. Same as .SALL.

	.LISTIF


	Starts listing of statements in false conditional blocks. Same as .LFCOND.
	.NOLISTIF


	Suppresses listing of conditional blocks whose condition evaluates to false (0). This is the default. Same as .SFCOND.

	.LISTALL


	Starts listing of all statements. Equivalent to the combination of .LIST, .LISTIF, and .LISTMACROALL.
	
	


